ORANI-G Model

Introduction to CGE Modeling Training

Arief Anshory Yusuf

Chiang Mai University, Thailand

March 9, 2023

The ORANI-G Model

- ORANI-G is an applied general equilibrium model which has been applied to many countries. It is descended from the ORANI GE model of the Australian economy which was used extensively for policy analysis in Australia for nearly two decades.
- ORANI-G (the 'G' stands for 'generic') is a version of ORANI designed both for teaching purposes and to serve as a basis from which to construct new models. Adaptations exist for many countries, including China, Thailand, South Africa, Korea, Pakistan, Brazil, the Philippines, Japan, Ireland, Vietnam, Indonesia, Venezuela, Taiwan and Denmark.
- Useful **website** about ORANI-G Model
- **Documentation** of ORANI-G model.

Worldwide application of ORANI-G



Figure: Countries shaded red are modelled by variants of the ORANI-G model, source: https://www.copsmodels.com/oranig.htm

Recall our convention!

CES production fn. with technological change:

$$Q = \frac{1}{A_Q}\left\{\sum_i \delta_i \left(\frac{X_i}{A_i}\right)^{\frac{\sigma_Q - 1}{\sigma_Q}}\right\}^{\frac{\sigma_Q}{\sigma_Q - 1}}$$

We can simplify this to:

$$Q\left(\frac{\mathbf{X_i}}{\mathbf{A_i}}; \sigma_Q, A_Q\right)$$

So, for example: $X = \frac{1}{A_X}\left(\delta_K \cdot \frac{K}{A_K}^{\frac{\sigma_X - 1}{\sigma_X}} + \delta_L \cdot \frac{L}{A_L}^{\frac{\sigma_X - 1}{\sigma_X}}\right)^{\frac{\sigma_X}{\sigma_X - 1}}$

can be written as $X\left(\frac{K}{A_K}, \frac{L}{A_L}; \sigma_X, A_X\right)$.

ORANI-G vs MINIMAL (separable) Production Function

MINIMAL production function:

$$\boxed{X_i\left(V_i\left(K_i, L_i; \sigma_V, A_V\right), \mathbf{Z_{ci}}; 0\right)}$$

ORANI-G production function:

$$\boxed{X_i\left(V_i\left(\frac{K_i}{A_i^K}, L_i\left(\mathbf{L^S_{io}}; \sigma_L, A_i^L\right), \frac{N_i}{A_i^N}; \sigma_V, A_i^V\right), \frac{\mathbf{Z_{ci}}}{\mathbf{A^Z_{ci}}}; 0, A_i^X\right)}$$

ORANI-G vs MINIMAL Model: Nested Production Function

ORANI-G model:

$$X_i\left(V_i\left(\frac{K_i}{A_i^K}, L_i\left(\mathbf{L^S_{io}};\sigma_L, A_i^L\right), \frac{N_i}{A_i^N};\sigma_V, A_i^V\right), \frac{\mathbf{Z_{ci}}}{\mathbf{A^Z_{ci}}};0, A_i^X\right)$$



MINIMAL model:

$$X_i\left(V_i\left(K_i, L_i;\sigma_V, A_V\right), \mathbf{Z_{ci}};0\right)$$



ORANI-G: Production

ORANI-G model's CES *separable* production function:

$$\boxed{X_i\left(V_i\left(\frac{K_i}{A_i^K}, L_i\left(\mathbf{L_{io}^S}; \sigma_L, A_i^L\right), \frac{N_i}{A_i^N}; \sigma_V, A_i^V\right), \frac{\mathbf{Z_{ci}}}{\mathbf{A_{ci}^Z}}; 0, A_i^X\right)}$$

With optimization and linearization:

- $k_i - a_i^K = v_i - \sigma_V\left(r_i + a_i^K - c_i\right)$
- $l_i - a_i^L = v_i - \sigma_V\left(w_i + a_i^L - c_i\right)$
- $n_i - a_i^N = v_i - \sigma_V\left(d_i + a_i^N - c_i\right)$
- $c_i = S_K \cdot r_i + S_L \cdot w + S_N \cdot d$
- $v_i - \left(a_i^V + a_i^X\right) = x_i$
- $z_{ci} - \left(a_{ci}^Z + a_i^X\right) = x_i$
- $l_{io}^S = l_i - \sigma_L\left(w_{io}^S - w_i\right)$
- $w_i = \sum_o S_{io}^{LS} w_{io}^S$

Production Function Tablo Equations - 1


```
E_x1lab_o  # Industry demands for effective labour #
  (all,i,IND)  x1lab_o(i) - a1lab_o(i) =
   x1prim(i) - SIGMA1PRIM(i)*[p1lab_o(i) + a1lab_o(i) - p1prim(i)];

E_p1cap  # Industry demands for capital #
  (all,i,IND)  x1cap(i) - a1cap(i) =
   x1prim(i) - SIGMA1PRIM(i)*[p1cap(i) + a1cap(i) - p1prim(i)];

E_p1lnd  # Industry demands for land #
  (all,i,IND)  x1lnd(i) - a1lnd(i) =
   x1prim(i) - SIGMA1PRIM(i)*[p1lnd(i) + a1lnd(i) - p1prim(i)];

E_p1prim  # Effective price term for factor demand equations #
  (all,i,IND)  V1PRIM(i)*p1prim(i) = V1LAB_O(i)*[p1lab_o(i) + a1lab_o(i)]
    + V1CAP(i)*[p1cap(i) + a1cap(i)] + V1LND(i)*[p1lnd(i) + a1lnd(i)];
```

Production Function Tablo Equations - 2


```
E_x1_s  # Demands for commodity composites #
 (all,c,COM)(all,i,IND)  x1_s(c,i) - [a1_s(c,i) + a1tot(i)] = x1tot(i);

E_x1prim  # Demands for primary factor composite #
 (all,i,IND)  x1prim(i) - [a1prim(i) + a1tot(i)] = x1tot(i);

 E_x1lab   # Demand for labour by industry and skill group #
  (all,i,IND)(all,o,OCC)
   x1lab(i,o) = x1lab_o(i) - SIGMA1LAB(i)*[p1lab(i,o) - p1lab_o(i)];

 E_p1lab_o # Price to each industry of labour composite #
  (all,i,IND) [TINY+V1LAB_O(i)]*p1lab_o(i)
                             = sum{o,OCC, V1LAB(i,o)*p1lab(i,o)};
```

ORANI-G Demand for Commodities

Demand aggregation structure is the same as MINIMAL model except investment is distinguished by industry:

Demand for domestic commodities:

$$Q_c^D = \sum_i Z_{ci}^D + H_c^D + \sum_i I_{ci}^D + G_c^D + E_c$$

Demand for imported commodities:

$$Q_c^M = \sum_i Z_{ci}^M + H_c^M + \sum_i I_{ci}^M + G_c^M$$

Equation for Aggregation of Total Demand

Tablo equations for the following:

$$Q_c^D = \sum_i Z_{ci}^D + H_c^D + \sum_i I_{ci}^D + G_c^D + E_c$$

$$Q_c^M = \sum_i Z_{ci}^M + H_c^M + \sum_i I_{ci}^M + G_c^M$$

```
Equation
E_delSaleA (all,c,COM)(all,s,SRC) delSale(c,s,"Interm") =
                   0.01*sum{i,IND,V1BAS(c,s,i)*x1(c,s,i)};
E_delSaleB (all,c,COM)(all,s,SRC) delSale(c,s,"Invest") =
                   0.01*sum{i,IND,V2BAS(c,s,i)*x2(c,s,i)};
E_delSaleC (all,c,COM)(all,s,SRC) delSale(c,s,"HouseH")=0.01*V3BAS(c,s)*x3(c,s);
E_delSaleD (all,c,COM)            delSale(c,"dom","Export")=0.01*V4BAS(c)*x4(c);
E_delSaleE (all,c,COM)            delSale(c,"imp","Export")=   0;
E_delSaleF (all,c,COM)(all,s,SRC) delSale(c,s,"GovGE") =0.01*V5BAS(c,s)*x5(c,s);
E_delSaleG (all,c,COM)(all,s,SRC) delSale(c,s,"Stocks") = LEVP0(c,s)*delx6(c,s);
```

Domestic-Import Sourcing

Sourcing principle is the same as MINIMAL but with different Tablo equations for each users (see next slide). *Note: D stands for domestic, and M stands for import.*

$$\text{minimize } P^D Q^D + P^M Q^M \text{ subject to } Q\left(Q^D, Q^M; \sigma_{DM}\right)$$

The (linearized) solution:

$$\boxed{q^D = q - \sigma_{DM}\left(p^D - p\right),\ q^M = q - \sigma_{DM}\left(p^M - p\right) \text{ and } p = S^D p^D + S^M p^M}$$

σ_{DM} is called **Armington elasticities**.

Domestic-Import Sourcing Equations


```
Equation E_x1  # Source-specific commodity demands #
 (all,c,COM)(all,s,SRC)(all,i,IND)
  x1(c,s,i)-a1(c,s,i) = x1_s(c,i) -SIGMA1(c)*[p1(c,s,i) +a1(c,s,i) -p1_s(c,i)];

Equation E_p1_s  # Effective price of commodity composite #
 (all,c,COM)(all,i,IND)
  p1_s(c,i) = sum{s,SRC, S1(c,s,i)*[p1(c,s,i) + a1(c,s,i)]};

Equation E_x2  # Source-specific commodity demands #
(all,c,COM)(all,s,SRC)(all,i,IND)
x2(c,s,i)-a2(c,s,i) - x2_s(c,i) = - SIGMA2(c)*[p2(c,s,i)+a2(c,s,i) - p2_s(c,i)];

Equation E_p2_s  # Effective price of commodity composite #
(all,c,COM)(all,i,IND)
p2_s(c,i) = sum{s,SRC, S2(c,s,i)*[p2(c,s,i)+a2(c,s,i)]};

Equation E_x3  # Source-specific commodity demands #
(all,c,COM)(all,s,SRC)
x3(c,s)-a3(c,s) = x3_s(c) - SIGMA3(c)*[ p3(c,s)+a3(c,s) - p3_s(c) ];

Equation E_p3_s  # Effective price of commodity composite #
(all,c,COM) p3_s(c) = sum{s,SRC, S3(c,s)*[p3(c,s)+a3(c,s)]};
```

Cobb-Douglas Demand System in 2x2 and MINIMAL

$$Q_i = \alpha_i \frac{Y}{P_i} \text{ for } i = 1, \ldots, n$$

or in linearized form:

$$q_i = y - p_i$$

Properties:

- Unitary income elasticity: $\frac{\partial Q_i}{\partial Y} \frac{Y}{Q_i} = 1$
- Unitary own-price elasticity: $\frac{\partial Q_i}{\partial P_i} \frac{P_i}{Q_i} = -1$
- Zero cross-price elasticity: $\frac{\partial Q_i}{\partial P_j} \frac{P_j}{Q_i} = 0$
- Constant budget share: $\frac{P_i Q_i}{Y} = \alpha_i$

Stone-Geary utility function

$$U = \prod_i (Q_i - \gamma_i)^{\beta_i}$$

- β_i is the marginal budget share and $\sum_i \beta_i = 1$
- γ_i is called the subsistence consumption level.
- $Q_i - \gamma_i$ is called supernumary consumption or luxurious consumption, the consumption after subsistence consumption
- Note that when $\gamma_i = 0$ the Stone-Geary utility function is collapsed to Cobb-Douglas
- See handout for derivation of LES Demand function

Some feature of the LES demand function:

$$Q_i = \gamma_i + \frac{\beta_i}{P_i}\left(Y - \sum_j P_j\gamma_j\right)$$

- Income elasticity: $\eta_i = \frac{\partial Q_i}{\partial Y}\frac{Y}{Q_i} = \frac{\beta_i}{P_i}\frac{Y}{Q_i} = \frac{\beta_i}{P_i}\frac{Y}{Q_i} = \frac{\beta_i}{w_i}$where $w_i = \frac{P_i Q_i}{Y}$ is the budget share.
- Own price elasticity: $\varepsilon_{ii} = \frac{\gamma_i(1-\beta_i)}{x_i} - 1$
- Cross-price elasticity: $\varepsilon_{ij} = -\frac{\beta_i\gamma_j}{x_j}\frac{w_j}{w_i}$
- Frisch parameter: $\Phi = -\frac{Y}{Y-\sum_j P_j\gamma_j}$, is the elasticity of marginal utility of income.

LES Household Demand in ORANI-G Tablo File

```
E_x3sub # Subsistence demand for composite commodities #
 (all,c,COM) x3sub(c) = q + a3sub(c);
E_x3lux # Luxury demand for composite commodities #
 (all,c,COM) x3lux(c) + p3_s(c) = w3lux + a3lux(c);
E_x3_s  # Total household demand for composite commodities #
 (all,c,COM) x3_s(c) = B3LUX(c)*x3lux(c) + [1-B3LUX(c)]*x3sub(c);
E_utility # Change in utility disregarding taste change terms #
 utility + q = sum{c,COM, S3LUX(c)*x3lux(c)};
E_a3lux # Default setting for luxury taste shifter #
 (all,c,COM) a3lux(c) = a3sub(c) - sum{k,COM, S3LUX(k)*a3sub(k)};
E_a3sub # Default setting for subsistence taste shifter #
 (all,c,COM) a3sub(c) = a3_s(c) - sum{k,COM, S3_S(k)*a3_s(k)};
E_w3tot # Household budget constraint: determines w3lux #
 w3tot = x3tot + p3tot;
```

Investment Demand

$$I_i^T = I_i^T \left(\mathbf{I^S_{ci}} \left(\frac{I_{ci}^D}{A_{ci}^D}, \frac{I_{ci}^M}{A_{ci}^M} ; \sigma^{ARM}, A_{ci}^S \right) ; 0, A_i^T \right)$$



Investment Demand Equation

$$\boxed{I_i^T\left(\mathbf{I^S_{ci}}\left(\frac{I^D_{ci}}{A^D_{ci}},\frac{I^M_{ci}}{A^M_{ci}};\sigma^{ARM},A^S_{ci}\right);0,A^T_i\right)}$$

With optimization and linearization:

- $i^D_{ci}-a^D_{ci}=i^S_{ci}-\sigma^{ARM}\left(p^D_{ci}+a^D_{ci}-p^S_{ci}\right)$
- $i^M_{ci}-a^M_{ci}=i^S_{ci}-\sigma^{ARM}\left(p^M_{ci}+a^M_{ci}-p^S_{ci}\right)$
- $p^S_{ci}=S^D_{ci}\left(p^D_{ci}-a^D_{ci}\right)+S^M_{ci}\left(p^M_{ci}-a^M_{ci}\right)$
- $i^S_{ci}-\left(a^S_{ci}+a^T_i\right)=i^T_i$
- $p^T_i=\sum_c S^I_{ci}\left(p^S_{ci}+a^S_{ci}+a^T_i\right)$

Investment Demand Equation in TABLO


```
E_x2  # Source-specific commodity demands #
(all,c,COM)(all,s,SRC)(all,i,IND)
x2(c,s,i)-a2(c,s,i) - x2_s(c,i) = - SIGMA2(c)*[p2(c,s,i)+a2(c,s,i) - p2_s(c,i)];

E_p2_s  # Effective price of commodity composite #
(all,c,COM)(all,i,IND)
p2_s(c,i) = sum{s,SRC, S2(c,s,i)*[p2(c,s,i)+a2(c,s,i)]};

E_x2_s  (all,c,COM)(all,i,IND)  x2_s(c,i) - [a2_s(c,i) + a2tot(i)] = x2tot(i);

E_p2tot (all,i,IND) p2tot(i)
  = sum{c,COM, (V2PUR_S(c,i)/ID01[V2TOT(i)])*[p2_s(c,i) +a2_s(c,i) +a2tot(i)]};
```

Investment allocation across industries: I_i^T

Investment in each industry can follow one of the following rules/behaviour.

1. Investment in one industry follows the profitability of the industry. Profitability is measured by the gross rate of returns of the industry calculated as the ratio of the rental of one unit capital to the price of acquiring new unit of capital (price of investment per unit), or:

$$\frac{I_i}{K_i} = F\left(\frac{R_i}{Pi^I}\right)$$

2. Investment in one industry follows the aggregate (macro-wide) investment. In the case where aggregate investment is exogenous, the investment in that industry will also be exogenous.
3. Investment to kapital ratio $\frac{I_i}{K_i}$ is exogenous (called long-run investment rule).

Investment allocation equation in TABLO


```
Variable
 (all,i,IND) ggro(i) # Gross growth rate of capital = Investment/capital #;
 (all,i,IND) gret(i) # Gross rate of return = Rental/[Price of new capital] #;
 (all,i,IND)  finv1(i)# Shifter to enforce DPSV investment rule #;
 (all,i,IND)  finv2(i)# Shifter for "exogenous" investment rule #;
 (all,i,IND)  finv3(i)# Shifter for longrun investment rule #;
 invslack # Investment slack variable for exogenizing aggregate investment #;

Equation
 E_ggro (all,i,IND) ggro(i) = x2tot(i) - x1cap(i);
 E_gret (all,i,IND) gret(i) = p1cap(i) - p2tot(i);

Equation E_finv1 # DPSV investment rule #
 (all,i,IND) ggro(i) = finv1(i) + 0.33*[2.0*gret(i) - invslack];

Equation E_finv2 # Alternative rule for "exogenous" investment industries #
 (all,i,IND) x2tot(i) = x2tot_i + finv2(i);

Equation E_finv3 # Alternative long-run investment rule #
 (all,i,IND) ggro(i) = finv3(i) + invslack;
```

Demand for Composite Commodities and Exports

- Demand of intermediate from industries is Leontief:

$$z_{ci} - \left(a_{ci}^{Z} + a_{i}^{X}\right) = x_i$$

- Demand for government is **exogenous** or follow household consumption.
- Demand for exports:

$$e_c = f_c^E - \epsilon_c \left(\left(p_c^D - \pi\right) - p_c^W\right)$$

where f_c^E is export shifter, ϵ_c is export elasticity, π is exhange rate, and p_c^W is international price.

Other composite demand in TABLO


```
Equation E_x1_s  # Demands for commodity composites #
 (all,c,COM)(all,i,IND)  x1_s(c,i) - [a1_s(c,i) + a1tot(i)] = x1tot(i);

 E_x5 # Government demands # (all,c,COM)(all,s,SRC) x5(c,s) = f5(c,s) + f5tot;
 E_f5tot # Overall government demands shift # f5tot = x3tot + f5tot2;

 E_pf4 (all,c,COM) pf4(c) = p4(c) - phi;
 E_x4A  # Individual export demand functions #
(all,c,TRADEXP) x4(c) - f4q(c) = -ABS[EXP_ELAST(c)]*[pf4(c) - f4p(c)];

Equation E_X4B  # Collective export demand functions #
 (all,c,NTRADEXP) x4(c) - f4q(c) = x4_ntrad;
```

Zero Profit Conditions Equation


```
Equation
 E_delV1CST  (all,i,IND) delV1CST(i) = delV1PRIM(i) +
  sum{c,COM,sum{s,SRC,  0.01*V1PUR(c,s,i)*[p1(c,s,i) + x1(c,s,i)]}}
                     +  0.01*V1OCT(i)    *[p1oct(i)  + x1oct(i)];

 E_delV1PTX  (all,i,IND) delV1PTX(i) =
                       PTXRATE(i)*delV1CST(i)  + V1CST(i) * delPTXRATE(i);

 E_delV1TOT  (all,i,IND) delV1TOT(i) = delV1CST(i) + delV1PTX(i);

 E_p1tot     (all,i,IND) V1TOT(i)*[p1tot(i) + x1tot(i)] = 100*delV1TOT(i);
```

Commodities Market Clearing Equations


```
Equation
E_delSaleA (all,c,COM)(all,s,SRC) delSale(c,s,"Interm") =
                   0.01*sum{i,IND,V1BAS(c,s,i)*x1(c,s,i)};
E_delSaleB (all,c,COM)(all,s,SRC) delSale(c,s,"Invest") =
                   0.01*sum{i,IND,V2BAS(c,s,i)*x2(c,s,i)};
E_delSaleC (all,c,COM)(all,s,SRC) delSale(c,s,"HouseH")=0.01*V3BAS(c,s)*x3(c,s);
E_delSaleD (all,c,COM)            delSale(c,"dom","Export")=0.01*V4BAS(c)*x4(c);
E_delSaleE (all,c,COM)            delSale(c,"imp","Export")=   0;
E_delSaleF (all,c,COM)(all,s,SRC) delSale(c,s,"GovGE") =0.01*V5BAS(c,s)*x5(c,s);
E_delSaleG (all,c,COM)(all,s,SRC) delSale(c,s,"Stocks") = LEVP0(c,s)*delx6(c,s);

Equation E_p0A # Supply = Demand for domestic commodities #
(all,c,COM) 0.01*[TINY+DOMSALES(c)]*x0dom(c) =sum{u,LOCUSER,delSale(c,"dom",u)};

Variable (all,c,COM)  x0imp(c)        # Total supplies of imported goods #;
Equation E_x0imp # Import volumes #
 (all,c,COM) 0.01*[TINY+V0IMP(c)]*x0imp(c) = sum{u,LOCUSER,delSale(c,"imp",u)};
```

Factor market clearing

Labor market clearing:

$$L_o^S = \sum_i L_{io}$$

```
Equation
 E_x1lab_i # Demand equals supply for labour of each skill #
  (all,o,OCC) V1LAB_I(o)*x1lab_i(o) = sum{i,IND, V1LAB(i,o)*x1lab(i,o)};

 E_p1lab   # Flexible setting of money wages #
  (all,i,IND)(all,o,OCC)
    p1lab(i,o)= p3tot + f1lab_io + f1lab_o(i) + f1lab_i(o) + f1lab(i,o);

Variable (all,o,OCC) p1lab_i(o) # Average wage of occupation #;
Equation E_p1lab_i # Average wage of occupation #
  (all,o,OCC) V1LAB_I(o)*p1lab_i(o) = sum{i,IND, V1LAB(i,o)*p1lab(i,o)};
```

Short-run and Long-run Closures

Technical changes are exogenous: $\mathbf{a1cap_i}$, $\mathbf{a1lab\_o_i}$, $\mathbf{a1lnd_i}$, $\mathbf{a1prim_i}$, $\mathbf{a1tot_i}$, $\mathbf{a2tot_i}$
Land and capital are sector-specific: $\mathbf{x1lnd_i}$, $\mathbf{x1cap_i} \Leftrightarrow \mathbf{gret_i}$
Wage follows CPI: $\mathbf{realwage} \Leftrightarrow \mathbf{employ\_i}$
Rate of returns are sector-specific: **capslack**
Expenditure side of GDP are constrained:
$\mathbf{x3tot} \Leftrightarrow \mathbf{delB}$, $\mathbf{x2tot\_i} \Leftrightarrow \mathbf{invslack}$, $\mathbf{x5tot} \Leftrightarrow \mathbf{f5tot2}$, $\mathbf{f5_{cs}}$, $\mathbf{delx6_c}$
Tax rates are exogenous: $\mathbf{delPTXRATE_i}$, $\mathbf{f0tax_{sc}}$, $\mathbf{f1oct_i}$, $\mathbf{f1tax\_csi}$, $\mathbf{f2tax\_csi}$, $\mathbf{f3tax\_cs}$, $\mathbf{f4tax\_ntrad}$, $\mathbf{f4tax\_trad}$, $\mathbf{f5tax\_cs}$, $\mathbf{t0imp_c}$
Some invest. follows profit: $\mathbf{finv1_i} \Leftrightarrow \mathbf{finv3_i}$, Some follows aggregate: $\mathbf{finv2_i}$
Import price and export demand curve are fixed:
$\mathbf{pf0cif_c}$, $\mathbf{f4p_c}$, $\mathbf{f4q_c}$, $\mathbf{f4p\_ntrad}$, $\mathbf{f4q\_ntrad}$
Preferences are exogenous: $\mathbf{q}$, $\mathbf{a3\_s_c}$ and exchange rate is numeraire: **phi**

Appendix: Consumer optimization with Stone-Geary utility function (1)

Consumer maximizes $U = \prod_i (X_i - \gamma_i)^{\beta_i}$ subject to $Y = \sum_i P_i X_i$
The Lagrange equation: $\mathcal{L} = \prod_i (X_i - \gamma_i)^{\beta_i} + \lambda (Y - \sum_i P_i X_i)$
First order conditions:

$$\beta_i (X_i - \gamma_i)^{\beta_i - 1} \prod_{j \neq i} (X_j - \gamma_j)^{\beta_j} = \lambda P_i \Rightarrow \frac{\beta_i U}{X_i - \gamma_i} = \lambda P_i \text{ and } Y = \sum_i P_i X_i$$

$$\frac{\beta_i (X_j - \gamma_j)}{\beta_j (X_i - \gamma_i)} = \frac{P_i}{P_j} \Rightarrow P_j \beta_i X_j - P_j \beta_i \gamma_j = P_i \beta_j (X_i - \gamma_i)$$

$$P_j X_j = \frac{P_i \beta_j (X_i - \gamma_i) + P_j \beta_i \gamma_j}{\beta_i}$$

Appendix: Consumer optimization with Stone-Geary utility function (2)

$$Y = \sum_j P_j X_j = \sum_j \frac{P_i \beta_j (X_i - \gamma_i) + P_j \beta_i \gamma_j}{\beta_i} = \sum_j \frac{\beta_j P_i X_i - \beta_j P_i \gamma_i + \beta_i P_j \gamma_j}{\beta_i}$$

$$Y = \frac{\sum_j \beta_j P_i X_i - \sum_j \beta_j P_i \gamma_i + \sum_j \beta_i P_j \gamma_j}{\beta_i}$$

$$Y = \frac{P_i X_i \sum_j \beta_j - P_i \gamma_i \sum_j \beta_j + \beta_i \sum_j P_j \gamma_j}{\beta_i} = \frac{P_i X_i - P_i \gamma_i}{\beta_i} + \sum_j P_j \gamma_j$$

Appendix: Consumer optimization with Stone-Geary utility function (3)

$$\frac{P_iX_i - P_i\gamma_i}{\beta_i} = Y - \sum_j P_j\gamma_j \Rightarrow P_iX_i - P_i\gamma_i = \beta_i\left(Y - \sum_j P_j\gamma_j\right)$$

$$P_iX_i = P_i\gamma_i + \beta_i\left(Y - \sum_j P_j\gamma_j\right) \quad \text{in expenditure form}$$

$$X_i = \gamma_i + \frac{\beta_i}{P_i}\left(Y - \sum_j P_j\gamma_j\right) \quad \text{in demand form}$$